<commit_before>"============================================================================
"File:        haxe.vim
"Description: Syntax checking plugin for syntastic.vim
"Maintainer:  <NAME> <<EMAIL>>
"License:     This program is free software. It comes without any warranty,
"             to the extent permitted by applicable law. You can redistribute
"             it and/or modify it under the terms of the Do What The Fuck You
"             Want To Public License, Version 2, as published by Sam Hocevar.
"             See http://sam.zoy.org/wtfpl/COPYING for more details.
"
"============================================================================

if exists("g:loaded_syntastic_haxe_haxe_checker")
    finish
endif
let g:loaded_syntastic_haxe_haxe_checker=1

function! SyntaxCheckers_haxe_haxe_IsAvailable()
    return executable('haxe')
endfunction

" start in directory a:where and walk up the parent folders until it
" finds a file matching a:what; return path to that file
function! s:FindInParent(what, where)
    let here = fnamemodify(a:where, ':p')

    while !empty(here)
        let p = split(globpath(here, a:what), '\n')

        if !empty(p)
            return fnamemodify(p[0], ':p')
        elseif here == '/'
            break
        endif

        " we use ':h:h' rather than ':h' since ':p' adds a trailing '/'
        " if 'here' is a directory
        let here = fnamemodify(here, ':p:h:h')
    endwhile

    return ''
endfunction

function! SyntaxCheckers_haxe_haxe_GetLocList()
    let hxml = exists('b:vaxe_hxml') ? fnamemodify(b:vaxe_hxml, ':p') : s:FindInParent('*.hxml', expand('%:p:h'))

    if !empty(hxml)
        let makeprg = syntastic#makeprg#build({
            \ 'exe': 'haxe',
            \ 'fname': shellescape(fnameescape(fnamemodify(hxml, ':t'))),
            \ 'filetype': 'haxe',
            \ 'subchecker': 'haxe' })

        let errorformat = '%E%f:%l: characters %c-%*[0-9] : %m'

        return SyntasticMake({
            \ 'makeprg': makeprg,
            \ 'errorformat': errorformat,
            \ 'cwd': fnamemodify(hxml, ':h') })
    endif

    return []
endfunction

call g:SyntasticRegistry.CreateAndRegisterChecker({
    \ 'filetype': 'haxe',
    \ 'name': 'haxe'})
<commit_msg>Make the haxe checker aware of g:vaxe_hxml.<commit_after>"============================================================================
"File:        haxe.vim
"Description: Syntax checking plugin for syntastic.vim
"Maintainer:  <NAME> <<EMAIL> at gmail dot com>
"License:     This program is free software. It comes without any warranty,
"             to the extent permitted by applicable law. You can redistribute
"             it and/or modify it under the terms of the Do What The Fuck You
"             Want To Public License, Version 2, as published by Sam Hocevar.
"             See http://sam.zoy.org/wtfpl/COPYING for more details.
"
"============================================================================

if exists("g:loaded_syntastic_haxe_haxe_checker")
    finish
endif
let g:loaded_syntastic_haxe_haxe_checker=1

function! SyntaxCheckers_haxe_haxe_IsAvailable()
    return executable('haxe')
endfunction

" start in directory a:where and walk up the parent folders until it
" finds a file matching a:what; return path to that file
function! s:FindInParent(what, where)
    let here = fnamemodify(a:where, ':p')

    while !empty(here)
        let p = split(globpath(here, a:what), '\n')

        if !empty(p)
            return fnamemodify(p[0], ':p')
        elseif here == '/'
            break
        endif

        " we use ':h:h' rather than ':h' since ':p' adds a trailing '/'
        " if 'here' is a directory
        let here = fnamemodify(here, ':p:h:h')
    endwhile

    return ''
endfunction

function! SyntaxCheckers_haxe_haxe_GetLocList()
    if exists('b:vaxe_hxml')
        let hxml = b:vaxe_hxml
    elseif exists('g:vaxe_hxml')
        let hxml = g:vaxe_hxml
    else
        let hxml = s:FindInParent('*.hxml', expand('%:p:h'))
    endif
    let hxml = fnamemodify(hxml, ':p')

    if !empty(hxml)
        let makeprg = syntastic#makeprg#build({
            \ 'exe': 'haxe',
            \ 'fname': shellescape(fnameescape(fnamemodify(hxml, ':t'))),
            \ 'filetype': 'haxe',
            \ 'subchecker': 'haxe' })

        let errorformat = '%E%f:%l: characters %c-%*[0-9] : %m'

        return SyntasticMake({
            \ 'makeprg': makeprg,
            \ 'errorformat': errorformat,
            \ 'cwd': fnamemodify(hxml, ':h') })
    endif

    return []
endfunction

call g:SyntasticRegistry.CreateAndRegisterChecker({
    \ 'filetype': 'haxe',
    \ 'name': 'haxe'})

{'fraction_non_alphanumeric': 0.18324046292327476, 'fraction_numerical': 0.0021431633090441492, 'mean_word_length': 2.921848739495798, 'pattern_counts': {'":': 0, '<': 9, '<?xml version=': 0, '>': 9, 'https://': 0, 'lorem ipsum': 0, 'www.': 0, 'xml': 21}, 'pii_count': 0, 'substrings_counts': 0, 'word_list_counts': {'cursed_substrings.json': 0, 'profanity_word_list.json': 0, 'sexual_word_list.json': 0, 'zh_pornsignals.json': 0}}
{'dir': 'git-commits-cleaned', 'id': '1052360', 'n_tokens_mistral': 1561, 'n_tokens_neox': 1499, 'n_words': 401}